Journal of Software for Algebra and Geometry

Computing quasidegrees of A-graded modules

Roberto Barrera

vol 9 2019

JSAG **9** (2019), 29–34
dx.doi.org/10.2140/jsag.2019.9.29

The Journal of Software for
Algebra and Geometry

Computing quasidegrees of A-graded modules

ROBERTO BARRERA

ABSTRACT: We describe the main functions of the Macaulay2 package `Quasidegrees.m2`. The purpose of this package is to compute the quasidegree set of a finitely generated $\mathbb{Z}^d$-graded module presented as the cokernel of a monomial matrix. We provide examples with motivation coming from A-hypergeometric systems.

1. INTRODUCTION. Throughout, $R = \Bbbk[x_1, \dots, x_n]$ is a $\mathbb{Z}^d$-graded polynomial ring over a field $\Bbbk$ and $\mathfrak{m} = \langle x_1, \dots, x_n \rangle$ denotes the homogeneous maximal ideal in R. Let $M = \bigoplus_{\beta \in \mathbb{Z}^d} M_\beta$ be a $\mathbb{Z}^d$-graded R-module. The *true degree set* of M is

$$\operatorname{tdeg}(M) = \{\beta \in \mathbb{Z}^d \mid M_\beta \neq 0\}.$$

The *quasidegree set* of M, denoted $\operatorname{qdeg}(M)$, is the Zariski closure in $\mathbb{C}^d$ of $\operatorname{tdeg}(M)$.

The purpose of the Macaulay2 package `Quasidegrees.m2` (provided as an online supplement) is to compute the quasidegree set of a finitely generated $\mathbb{Z}^d$-graded module presented as the cokernel of a monomial matrix. By a monomial matrix, we mean a matrix where each entry is either zero or a monomial in R. The initial motivation for `Quasidegrees.m2` was to compute the quasidegree sets of certain local cohomology modules supported at $\mathfrak{m}$ of $\mathbb{Z}^d$-graded R-modules, so there are some methods in the package specific to local cohomology. Recall that the *i-th local cohomology module* of M with support at the ideal $I \subset R$ is the i-th right derived functor of the left exact I-torsion functor

$$\Gamma_I(M) = \{m \in M \mid I^t m = 0 \text{ for some } t \in \mathbb{N}\}$$

on the category of R-modules.

By the vanishing theorems of local cohomology [Eisenbud 1995], the quasidegree sets of the local cohomology modules supported at $\mathfrak{m}$ of M can be seen as measuring how far the module is from being Cohen–Macaulay. From the A-hypergeometric systems point of view, the quasidegree set of the non-top local cohomology modules supported at $\mathfrak{m}$ of R/I_A, where I_A is the toric ideal associated

MSC2010: 13D45, 13P20, 33C70.
Keywords: quasidegree, hypergeometric system, local cohomology.
`Quasidegrees.m2` version 1.0

to A in R, determine the parameters β where the A-hypergeometric system $H_A(\beta)$ has rank higher than expected (see Section 3).

2. QUASIDEGREES. The main function of `Quasidegrees.m2` is `quasidegrees`, which computes the quasidegree set of a module that is presented by a monomial matrix.

We use the idea of standard pairs of monomial ideals to compute the quasidegree set of a $\mathbb{Z}^d$-graded R-module. Given a monomial $x^{\boldsymbol{u}}$ and a subset $Z \subset \{x_1, \dots, x_n\}$, the pair $(x^{\boldsymbol{u}}, Z)$ indexes the monomials $x^{\boldsymbol{u}} \cdot x^{\boldsymbol{v}}$ where $\operatorname{supp}(x^{\boldsymbol{v}}) \subset Z$. A *standard pair* of a monomial ideal $I \subset R$ is a pair $(x^{\boldsymbol{u}}, Z)$ satisfying:

(1) $\operatorname{supp}(x^{\boldsymbol{u}}) \cap Z = \varnothing$.

(2) All of the monomials indexed by $(x^{\boldsymbol{u}}, Z)$ are outside of I.

(3) $(x^{\boldsymbol{u}}, Z)$ is maximal in the sense that $(x^{\boldsymbol{u}}, Z) \not\subseteq (x^{\boldsymbol{v}}, Y)$ for any other pair $(x^{\boldsymbol{v}}, Y)$ satisfying the first two conditions.

To compute the quasidegree set of M we first find a monomial presentation of M so that M is the cokernel of a monomial matrix ϕ. We then compute the standard pairs of the ideals generated by the rows of ϕ and to each standard pair we associate the degrees of the corresponding variables. Algorithm 1 below is implemented in `Quasidegrees.m2`. The input is an R-module presented by a monomial matrix

$$\phi : R^s \to R^t.$$

As in Macaulay2, we write the degree of the k-th factor of R^t next to the k-th row of the matrix ϕ.

In the Macaulay2 implementation of the algorithm, we represent the output as a list of pairs $(\boldsymbol{u}, Z)$ with $\boldsymbol{u} \in \mathbb{Q}^d$ and $Z \subset \mathbb{Q}^d$, where the pair $(\boldsymbol{u}, Z)$ represents the plane

$$\boldsymbol{u} + \sum_{\boldsymbol{v} \in Z} \mathbb{C} \cdot \boldsymbol{v}.$$

Input: R-module M presented by monomial matrix $\phi = \alpha_i [c_{j,k} \boldsymbol{x}^{\boldsymbol{u}_{j,k}}] : R^s \to R^t$
Output: $\operatorname{qdeg}(M)$

$Q = \varnothing$
for $1 \le k \le t$ **do**
  $SP = \{\text{standard pairs of } \langle c_{k,1}\boldsymbol{x}^{\boldsymbol{u}_{k,1}}, c_{k,2}\boldsymbol{x}^{\boldsymbol{u}_{k,2}}, \dots, c_{k,s}\boldsymbol{x}^{\boldsymbol{u}_{k,s}} \rangle\}$
  $Q = Q \cup \{\deg(\boldsymbol{x}^{\boldsymbol{u}}) + \alpha_k + \sum_{x_i \in F} \mathbb{C} \cdot \deg(x_i) \mid (\boldsymbol{x}^{\boldsymbol{u}}, Z) \in SP\}$
end for
return Q

Algorithm 1. Compute $\operatorname{qdeg}(M)$.

The union of these planes over all such pairs in the output is the quasidegree set of M.

The following is an example of `Quasidegrees.m2` computing the quasidegree set of an R-module:

```
i1 : R=QQ[x,y,z]
o1 = R
o1 : PolynomialRing
i2 : I=ideal(x*y,y*z)
o2 = ideal (x*y, y*z)
o2 : Ideal of R
i3 : M=R^1/I
o3 = cokernel | xy yz |
                          1
o3 : R-module, quotient of R
i4 : Q = quasidegrees M
o4 = {{0, {| 1 |}}, {0, {| 1 |, | 1 |}}}
o4 : List
```

The above example displays a caveat of `quasidegrees` in that there may be some redundancies in the output. By a redundancy, we mean when one plane in the output is contained in another. The redundancy above is clear:

$$\operatorname{qdeg}(\Bbbk[x, y, z]/\langle xy, yz\rangle) = \mathbb{C} = \{z_1 + z_2 \in \mathbb{C} \mid z_1, z_2 \in \mathbb{C}\}.$$

The function `removeRedundancy` gets rid of redundancies in the list of planes:

```
i5 : removeRedundancy Q
o5 = {{0, {| 1 |, | 1 |}}}
o5 : List
```

3. Quasidegrees and hypergeometric systems. In this section, we discuss the motivation for `Quasidegrees.m2` and the methods therein which aid us in our studies. Let $A = [a_1\ a_2\ \cdots\ a_n]$ be an integer $(d \times n)$-matrix with $\mathbb{Z}A = \mathbb{Z}^d$ and such that the cone over its columns is pointed. There is a natural $\mathbb{Z}^d$-grading of R by the columns of A given by $\deg(x_j) = a_j$, the j-th column of A. A module that is homogeneous with respect to this grading is said to be *A-graded*. By the assumptions on A, R is positively graded by A, that is, the only polynomials of degree 0 are the constants. Given such a matrix A and a polynomial ring R in n variables, the method `toGradedRing` gives R an A-grading. For example, let

$$A = \begin{pmatrix} 1 & 1 & 1 & 1 & 1 \\ 0 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & -2 \end{pmatrix}.$$

We make the A-graded polynomial ring $\mathbb{Q}[x_1, x_2, x_3, x_4, x_5]$:

```
i6 : A=matrix{{1,1,1,1,1},{0,0,1,1,0},{0,1,1,0,-2}}
o6 = | 1 1 1 1 1  |
     | 0 0 1 1 0  |
     | 0 1 1 0 -2 |
              3        5
o6 : Matrix ZZ  <--- ZZ
i7 : R=QQ[x_1..x_5]
o7 = R
o7 : PolynomialRing
i8 : R=toGradedRing(A,R)
o8 = R
o8 : PolynomialRing
i9 : describe R
o9 = QQ[x , x , x , x , x , Degrees => {{1}, {1}, {1}, {1}, {1 }},
         1   2   3   4   5              {0}  {0}  {1}  {1}  {0 }
                                        {0}  {1}  {1}  {0}  {-2}
     Heft=>{1, 2:0},MonomialOrder=>{MonomialSize=>32},DegreeRank=>3]
                                   {GRevLex=>{5:1}}
                                   {Position=>Up}
```

The *toric ideal associated to* A in R is the binomial ideal

$$I_A = \langle \boldsymbol{x}^{\boldsymbol{u}} - \boldsymbol{x}^{\boldsymbol{v}} : A\boldsymbol{u} = A\boldsymbol{v} \rangle.$$

The method `toricIdeal` computes the toric ideal associated to A in the ring R. We continue with the A and R from the above example and compute the toric ideal I_A associated to A in R:

```
i10 : I=toricIdeal(A,R)
                                  2     2     2             3    2
o10 = ideal (x x  - x x , x x  - x x , x x  - x x x , x  - x x )
              1 3    2 4   1 4    3 5   1 4    2 3 5   1    2 5
o10 : Ideal of R
```

We now introduce A-hypergeometric systems. Given a matrix $A \in \mathbb{Z}^{d\times n}$ as above and a $\beta \in \mathbb{C}^d$, the *A-hypergeometric system with parameter $\beta \in \mathbb{C}^d$* [Saito et al. 2000], denoted $H_A(\beta)$, is the system of partial differential equations:

$$\frac{\partial^{|\boldsymbol{v}|}}{\partial \boldsymbol{x}^{\boldsymbol{v}}}\phi(\boldsymbol{x}) = \frac{\partial^{|\boldsymbol{u}|}}{\partial \boldsymbol{x}^{\boldsymbol{u}}}\phi(\boldsymbol{x}) \qquad \text{for all } \boldsymbol{u}, \boldsymbol{v},\ A\boldsymbol{u} = A\boldsymbol{v},$$

$$\sum_{j=1}^{n} a_{ij} x_j \frac{\partial}{\partial x_j}\phi(\boldsymbol{x}) = \beta_i \phi(\boldsymbol{x}), \qquad \text{for } i = 1, \dots, d.$$

Such systems are sometimes called *GKZ-hypergeometric systems*. The function gkz in the Macaulay2 package Dmodules computes this system as an ideal in the Weyl algebra. The *rank* of $H_A(\beta)$ is

$$\operatorname{rank}(H_A(\beta)) = \dim_{\mathbb{C}} \left\{ \begin{array}{l} \text{germs of holomorphic solutions of } H_A(\beta) \\ \text{near a generic nonsingular point} \end{array} \right\}.$$

The function holonomicRank in Dmodules computes the rank of an A-hypergeometric system. In general, rank is not a constant function of β. Denote $\operatorname{vol}(A)$ to be $d!$ times the Euclidean volume of $\operatorname{conv}(A \cup \{0\})$, the convex hull of the columns of A and the origin in $\mathbb{R}^d$. The following theorem gives the parameters β for which $\operatorname{rank}(H_A(\beta))$ is higher than expected:

Theorem 3.1 [Matusevich et al. 2005]. *Let $H_A(\beta)$ be an A-hypergeometric system with parameter β. If $\beta \in \operatorname{qdeg}(\bigoplus_{i=0}^{d-1} H^i_{\mathfrak{m}}(R/I_A))$ then $\operatorname{rank}(H_A(\beta)) > \operatorname{vol}(A)$. Otherwise*, $\operatorname{rank}(H_A(\beta)) = \operatorname{vol}(A)$.

Since Theorem 3.1 was the initial motivation for Quasidegrees.m2, the package has a method quasidegreesLocalCohomology (abbreviated qlc) to compute the quasidegree set of the local cohomology modules $H^i_{\mathfrak{m}}(R/I_A)$. If the input is an integer i and the R-module R/I_A, then the method computes $\operatorname{qdeg}(H^i_{\mathfrak{m}}(R/I_A))$. If the input is only the module R/I_A, the method computes the quasidegree set in Theorem 3.1.

We use graded local duality to compute the local cohomology modules of a finitely generated A-graded R-module supported at the maximal ideal $\mathfrak{m}$:

Theorem 3.2 (graded local duality [Bruns and Herzog 1993; Miller 2002]). *Given an A-graded R-module M, there is an A-graded vector space isomorphism*

$$\operatorname{Ext}^{n-i}_R(M, R)_\alpha \cong \operatorname{Hom}_{\Bbbk}(H^i_{\mathfrak{m}}(M)_{-\alpha-\varepsilon_A}, \Bbbk),$$

where $\mathfrak{m} = \langle x_1, \dots, x_n \rangle$ and $\varepsilon_A = \sum_{j=1}^n a_j$.

The algorithm implemented for quasidegreesLocalCohomology is essentially Algorithm 1 applied to the Ext-modules of M with the additional twist of ε_A coming from local duality. For our purposes, we exploit the fact that the higher syzygies of R/I_A are generated by monomials in R^m (see [Miller and Sturmfels 2005], Chapter 9).

Continuing our running example, we use quasidegreesLocalCohomology to compute the quasidegree set of $\bigoplus_{i=0}^{d-1} H^i_{\mathfrak{m}}(R/I_A)$:

```
i11 : M=R^1/I
o11 = cokernel | x_1x_3-x_2x_4 x_1x_4^2-x_3^2x_5
x_1^2x_4-x_2x_3x_5 x_1^3-x_2^2x_5 |
                                    1
o11 : R-module, quotient of R
```

```
i12 : quasidegreesLocalCohomology M
o12 = {{| 0 |, {| 1  |}}}
        | 0 |    | 0  |
        | 1 |    | -2 |
o12 : List
```

Thus

$$\operatorname{qdeg}\Bigl(\bigoplus_{i=0}^{d-1} H^i_{\mathfrak{m}}(R/I_A)\Bigr) = \begin{bmatrix}0\\0\\1\end{bmatrix} + \mathbb{C}\cdot\begin{bmatrix}1\\0\\-2\end{bmatrix}. \tag{1}$$

As a check, we use the methods `gkz` and `holonomicRank` from the package `Dmodules` to compute $\operatorname{rank}(H_A(0))$ and $\operatorname{rank}(H_A(\beta))$ for two different β in (1) and demonstrate a rank jump:

```
i13 : holonomicRank gkz(A,{0,0,0}) -- vol A in this case
o13 = 4
i14 : holonomicRank gkz(A,{0,0,1})
o14 = 5
i15 : holonomicRank gkz(A,{3/2,0,-2})
o15 = 5
```

SUPPLEMENT. The online supplement contains version 1.0 of `Quasidegrees.m2`.

REFERENCES.

[Bruns and Herzog 1993] W. Bruns and J. Herzog, *Cohen–Macaulay rings*, Cambridge Studies in Advanced Mathematics **39**, Cambridge University Press, 1993. MR Zbl

[Eisenbud 1995] D. Eisenbud, *Commutative algebra: with a view toward algebraic geometry*, Graduate Texts in Mathematics **150**, Springer, 1995. MR Zbl

[Matusevich et al. 2005] L. F. Matusevich, E. Miller, and U. Walther, "Homological methods for hypergeometric families", *J. Amer. Math. Soc.* **18**:4 (2005), 919–941. MR Zbl

[Miller 2002] E. Miller, "Graded Greenlees–May duality and the Čech hull", pp. 233–253 in *Local cohomology and its applications* ((Guanajuato, 1999)), edited by G. Lyubeznik, Lecture Notes in Pure and Appl. Math. **226**, Dekker, New York, 2002. MR Zbl

[Miller and Sturmfels 2005] E. Miller and B. Sturmfels, *Combinatorial commutative algebra*, Graduate Texts in Mathematics **227**, Springer, 2005. MR Zbl

[Saito et al. 2000] M. Saito, B. Sturmfels, and N. Takayama, *Gröbner deformations of hypergeometric differential equations*, Algorithms and Computation in Mathematics **6**, Springer, 2000. MR Zbl

RECEIVED: 27 Jun 2015 REVISED: 13 Oct 2018 ACCEPTED: 26 Feb 2019

ROBERTO BARRERA:

rbarrera@txstate.edu
Department of Mathematics, Texas State University, San Marcos, TX, United States

msp

JOURNAL OF SOFTWARE FOR ALGEBRA AND GEOMETRY vol 9, no 1, 2019

1948-7916(2019)9:1;1-D